Line Integal with respect to arc length visual example

We wish to find the area of a sheet that is above the quarter unit circle in the first quadrant:

$$x^2 + y^2 = 1 \quad \text{where } x, y \geq 0$$

and is below the plane:

$$z = -1 + x + y$$

First, we'll plot a picture of this sheet and then we will calculate its area using a line integral (with respect to arc length).

```
> restart;
  with(plots):
  with(VectorCalculus):

  # For convenience, let's define the standard unit vectors...
  i := <1,0,0>;
  j := <0,1,0>;
  k := <0,0,1>;
```

$$i := (1)e_x + (0)e_y + (0)e_z$$

$$j := (0)e_x + (1)e_y + (0)e_z$$

$$k := (0)e_x + (0)e_y + (1)e_z \qquad \textbf{(1)}$$

```
> # This is our plane (the surface above our sheet)...
  f := (x,y) -> -1+x+y:
  'f(x,y)' = f(x,y);

  # This parameterizes the unit circle in the xy-plane...
  r := t -> <cos(t),sin(t),0>:
  'r(t)' = r(t);

  # We plot...
  # (1) The plane in somewhat transparent yellow.
  # (2) The quarter cricle in the xy-plane in thick blue.
  # (3) The sheet in green. This plot is done using a parameterization
  of the sheet
  #     surface. r(t).i and r(t).j pick off the i and j components of
  our curve's
  #     parameterization. Basically we have: <x(t), y(t), s * f(x(t),y
  (t))>
  #     Think of t traveling along the curve while s takes us up from 0
  to the
  #     height of the plane.
  # (4) The xy-plane in mostly transparent gray.
  #
  surfacePlot := plot3d(f(x,y),x=0..1,y=0..1,color=yellow,transparency=
  0.5):
  curvePlot := spacecurve(r(t),t=0..Pi/2,thickness=3,color=blue):
  sheetPlot := plot3d(<r(t).i,r(t).j,s*f(r(t).i,r(t).j)>,t=0..Pi/2,s=0.
  .1,color=green):
  xyPlanePlot := plot3d(0,x=0..1,y=0..1,color=gray,transparency=0.8):

  # Show all these plots together...
  display({surfacePlot, curvePlot, sheetPlot, xyPlanePlot},scaling=
```

```
constrained);
```

$$f(x,y) = -1 + x + y$$

$$r(t) = (\cos(t))e_x + (\sin(t))e_y + (0)e_z$$



Now we calculate the area.

```
> # ds is the length of the derivative of our paramterization.
  # This is the "arc length element"
  ds := simplify(Norm(diff(r(t),t)));

  # We plug x(t) = r(t).i in for x, y(t) = r(t).j in for y, and
  # |r'(t)| in for ds. The left hand side is "inert" to show off
  # what integral is being calculated.
  Int(f(r(t).i,r(t).j)*ds,t=0..Pi/2) = int(f(r(t).i,r(t).j)*ds,t=0..
  Pi/2);

  # A decimal approximation of the area...
  evalf(int(f(r(t).i,r(t).j)*ds,t=0..Pi/2));
```

$$ds := 1$$

$$\int_0^{\frac{\pi}{2}} (-1 + \cos(t) + \sin(t))\,dt = 2 - \frac{\pi}{2}$$

$$0.429203673 \qquad (2)$$